
* I) RECODING VARIABLES AND DATA MANAGEMENT

recode region (1=27131)(2=27132)(7=27133)(3=27134)(4=27181)(5=27183)(6=27184)(8=27185)(9=27191)(10=27192)(11=27193)(12=27194) ///
(13=27161)(14=27162)(15=27163)(16=27164)(17=27165)(18=27166)(19=27171)(20=27172), gen (region_code_n)

recode region (1/3 7 = 1 "N-west")(4 5 6 8 = 2 "N-east")(9 10 11 12 = 3 "Centre")(13/20=4 "South and Islands"), gen(macroarea)
fre macroarea
replace region_code_n = 27182 if region == 4 & provincia==52   /* Trento: Overall 21 subregional areas */

* Battery on vaccines (v289_it v290_it) and big pharma conspiracy theory (v291_it)
fre v289_it v290_it v291_it
recode v289_it v290_it v291_it (8 9=.), gen (vaxharm1 vaxtrust pharma1) 
pwcorr vaxharm1 vaxtrust pharma1
gen pharma = (6 - pharma1) 
gen vaxharm = (6 - vaxharm1) 
label define likert 1 "disagree strongly" 2 "disagree" 3 "neither agree nor disagree" 4 "agree" 5 "agree strongly"
label values pharma likert
label values vaxharm likert
label values vaxtrust v290_it
drop vaxharm1 pharma1
pwcorr vaxharm vaxtrust pharma

* Socio-demographics
fre v225h /*gender: 1-male 2-female */
fre v243rh /* educational level */
recode v243rh (-2=.), gen (edu3)
label define edu3 1"low" 2"medium" 3"high"
label values edu3 edu3
fre age6cat /* age 6 categories */

*** Religiosity

* Belonging
recode v51h (1=1 "Yes")(0=0 "No") (else=.), gen (rlgblg)

* Beliefs
recode v57h v58h v59h v60h (1=1)(0 -1 =0) (-2=.), gen (god life hell heaven)   /* considero i Sì vs No+Do not Know */
label define belief 1 "yes" 0 "no/dk"
label values god belief
label values life belief
label values hell belief
label values heaven belief
alpha god life hell heaven, std detail item
factor god life hell heaven, pcf mineigen(0.7)

gen believe4= (life+hell+heaven+god)/4

* Belief in Reincarnation
recode v61h (1=1)(0 -1 =0) (-2=.), gen (reincarnation)   /* considero i Sì vs No+Do not Know */

corr god life hell heaven reincarnation
corr god life hell heaven reincarnation
factor god life hell heaven reincarnation, pcf mineigen(0.7)
rotate, oblimin blank (0.6)

* Church attendance
recode v54h (1 2 =1 "Regular")(3 4 5 6 =2 "Irregular") (7 8 =3 "Non practising") (else=.) , gen (chatt3)


* Attitudes toward science
recode  v283_it v284_it v285_it v286_it v287_it v288_it (88 99 =.), gen(sclifeasier scoppor scfaith scconfond scknow sctecworldbetter) 
alpha sclifeasier scoppor scfaith scconfond scknow sctecworldbetter, item std

* TABLE S1
factor sclifeasier scoppor scfaith scconfond scknow, pcf mineigen (0.8)
rotate, varimax blank(0.5) 

* TABLE S2
factor sclifeasier scoppor scfaith scconfond, pcf mineigen (0.8)
rotate, varimax blank(0.5) 

alpha sclifeasier scoppor, item std
alpha scfaith scconfond scknow, item std /* * scknow (uniqueness >0.5) */
alpha scfaith scconfond, item std

gen scest = (sclifeasier + scoppor)/2
gen scint = (scfaith + scconfond)/2

* Interpersonal trust
fre v31h
recode v31h (2=0) (-1 -2 =.), gen (gentrust) 
label define gentrust 0 "can´t be too careful" 1 "most people can be trusted"
label values gentrust gentrust

* Confidence in institutions
recode v115h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (church)
recode v116h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (armedforces)
recode v117h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (edusyst)
recode v118h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (press)
recode v119h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (unions)
recode v120h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (police)
recode v121h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (parliament)
recode v122h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (civservice)
recode v123h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (socsecursyst)
recode v124h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (EU)
recode v125h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (ONU)
recode v126h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (healthsyst)
recode v127h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (justsyst)
recode v128h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (companies)
recode v129h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (envorg)
recode v130h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (parties)
recode v131h (-1 -2=.) (1=3 "a great deal")(2=2 "quite a lot")(3=1 "not very much")(4=0 "not at all"), gen (government)

alpha parliament parties justsyst civservice government, item std
gen poltrust2 = (parliament + parties + justsyst + civservice)/4


* II) FIGURES AND TABLES

* FIGURE 1
hist pharma, percent

**
reg pharma ib3.chatt3 c.believe4 i.reincarnation v225h i.edu3 i.age6cat gentrust poltrust2 church c.scint c.scest 

* TABLE 1, MODEL 1
reg pharma c.believe4 i.reincarnation if e(sample)

* TABLE 1, MODEL 2
reg pharma ib3.chatt3 c.believe4 i.reincarnation if e(sample)

* TABLE 1, MODEL 3
reg pharma ib3.chatt3 c.believe4 i.reincarnation v225h i.edu3 i.age6cat if e(sample)

* TABLE 1, MODEL 4
reg pharma ib3.chatt3 c.believe4 i.reincarnation v225h i.edu3 i.age6cat gentrust if e(sample)

* TABLE 1, MODEL 5
reg pharma ib3.chatt3 c.believe4 i.reincarnation v225h i.edu3 i.age6cat gentrust poltrust2 church if e(sample)

* TABLE 1, MODEL 6
reg pharma ib3.chatt3 c.believe4 i.reincarnation v225h i.edu3 i.age6cat gentrust poltrust2 church c.scint c.scest if e(sample)


* SYNTAX FOR TABLE A1
fre reincarnation v225h edu3 age6cat gentrust if e(sample)
sum believe4 poltrust2 church scint scest if e(sample)

* Dichotomizing the dependent variable
recode pharma (4 5 = 1)(1 2 3 = 0), gen (pharma2)

logit pharma2 ib3.chatt3 c.believe4 i.reincarnation v225h i.edu3 i.age6cat gentrust poltrust2 church c.scint c.scest 

* TABLE A2, MODEL 1
logit pharma2 c.believe4 i.reincarnation if e(sample)

* TABLE A2, MODEL 2
logit pharma2 ib3.chatt3 c.believe4 i.reincarnation if e(sample)

* TABLE A2, MODEL 3
logit pharma2 ib3.chatt3 c.believe4 i.reincarnation v225h i.edu3 i.age6cat if e(sample)

* TABLE A2, MODEL 4
logit pharma2 ib3.chatt3 c.believe4 i.reincarnation v225h i.edu3 i.age6cat gentrust if e(sample)

* TABLE A2, MODEL 5
logit pharma2 ib3.chatt3 c.believe4 i.reincarnation v225h i.edu3 i.age6cat gentrust poltrust2 church if e(sample) 

* TABLE A2, MODEL 6
logit pharma2 ib3.chatt3 c.believe4 i.reincarnation v225h i.edu3 i.age6cat gentrust poltrust2 church c.scint c.scest if e(sample)


